Solving a third order non-homogeneous linear differential equation with constant coefficients.

We will do this 5 ways: undetermined coefficients, variation of parameters, partial fractions trick, conversion to a first order linear system, and cheating and using Maple's dsolve.

```
> restart;
  with(LinearAlgebra):
  with(VectorCalculus):
> ? Wronskian
```

Let's cook up an equation with a couple interesting features...

```
> expand((x+3)*(x-1)^2);
```

$$x^3 + x^2 - 5x + 3 \quad (1)$$

```
> DEqn := diff(y(t),t,t,t)+diff(y(t),t,t)-5*diff(y(t),t)+3*y(t)=6*t+exp(-t)*
  sin(4*t)-7*t*exp(-3*t);
```

$$DEqn := \frac{d^3}{dt^3}\, y(t) + \frac{d^2}{dt^2}\, y(t) - 5\, \frac{d}{dt}\, y(t) + 3\, y(t) = 6\,t + e^{-t} \sin(4\,t) - 7\,t\,e^{-3\,t} \quad (2)$$

For our first few methods we begin by solving the **Companion Equation** (i.e., the homogeneous part):

```
> factor(s^3+s^2-5*s+3)=0;
```

$$(s+3)\,(s-1)^2 = 0 \quad (3)$$

```
> yh := C1*exp(-3*t)+C2*exp(t)+C3*t*exp(t);
```

$$yh := C1\,e^{-3\,t} + C2\,e^{t} + C3\,t\,e^{t} \quad (4)$$

Undetermined Coeffs:

Notice that 6t is annihilated by D^2 and $e^{-t}\sin(4\,t)$ is annihilated by $(D+1-4\,i)(D+1+4\,i)$ and $-7\,t\,e^{-3\,t}$ is annhilated by $(D+3)^2$. Notice that the final part overlaps with our companion equation. Thus as we construct our guess, we need to adjust the last piece by multiplying by t.

```
> yp := a*t+b+c*exp(-t)*sin(4*t)+d*exp(-t)*cos(4*t)+t*(e*t+f)*exp(-3*t);
```

$$yp := a\,t + b + c\,e^{-t}\sin(4\,t) + d\,e^{-t}\cos(4\,t) + t\,(e\,t+f)\,e^{-3\,t} \quad (5)$$

We then plug our guess into the differential equation and read off coefficients to get our (linear) equations determining the unknown coefficients:

```
> simplify(diff(yp,t,t,t)+diff(yp,t,t)-5*diff(yp,t)+3*yp
  =6*t+exp(-t)*sin(4*t)-7*t*exp(-3*t));
```

$$((-80\,c+40\,d)\cos(4\,t) + 40\sin(4\,t)\,(c+2\,d))\,e^{-t} + ((32\,t-16)\,e+16\,f)\,e^{-3\,t} + (3\,t-5)\,a + 3\,b = 6\,t + e^{-t}\sin(4\,t) - 7\,t\,e^{-3\,t} \quad (6)$$

```
> solve({40*c+80*d=1,-80*c+40*d=0,-16*e+16*f=0,16*e+16*e=-7,-5*a+3*b=0,3*a=6}
  );
```

(7)

$$\left\{a=2, b=\frac{10}{3}, c=\frac{1}{200}, d=\frac{1}{100}, e=-\frac{7}{32}, f=-\frac{7}{32}\right\} \quad (7)$$

```
> a := 2; b := 10/3; c := 1/200; d := 1/100; e := -7/32; f := -7/32;
```

$$a := 2$$
$$b := \frac{10}{3}$$
$$c := \frac{1}{200}$$
$$d := \frac{1}{100}$$
$$e := -\frac{7}{32}$$
$$f := -\frac{7}{32} \quad (8)$$

```
> yp;
```

$$2\,t+\frac{10}{3}+\frac{e^{-t}\sin(4\,t)}{200}+\frac{e^{-t}\cos(4\,t)}{100}+t\left(-\frac{7\,t}{32}-\frac{7}{32}\right)e^{-3\,t} \quad (9)$$

Check it really is a solution (this is left hand side of our DE minus g -- it should be zero):

```
> simplify(diff(yp,t,t,t)+diff(yp,t,t)-5*diff(yp,t)+3*yp-(6*t+exp(-t)*sin(4*
  t)-7*t*exp(-3*t)));
```

$$0 \quad (10)$$

We get the following general solution:

```
> y = yh+yp;
```

$$y=C1\,e^{-3\,t}+C2\,e^{t}+C3\,t\,e^{t}+2\,t+\frac{10}{3}+\frac{e^{-t}\sin(4\,t)}{200}+\frac{e^{-t}\cos(4\,t)}{100}+t\left(-\frac{7\,t}{32}-\frac{7}{32}\right)e^{-3\,t} \quad (11)$$

Variation of Parameters:

Here we build our particular solution yp from the homogeneous solution where constants are replaced by the v_i functions. Cramer's rule etc. tell us that v_i is the integral of $(-1)^{i+n}g(t)$ times the Wronskian of all but the i-th homogeneous solution then divided by the Wronskian of the whole solution set. [Here n=3 since this is a 3rd order equation.]

Note: Wronskian kicks out the Wronskian matrix. "determinant=true" makes it kick out both matrix and determinant, so the "[2]" picks off the desired Wronskian determinant.

```
> yp := v1*exp(-3*t)+v2*exp(t)+v3*t*exp(t);
  g := 6*t+exp(-t)*sin(4*t)-7*t*exp(-3*t);
```

$$yp := v1\,e^{-3\,t}+v2\,e^{t}+v3\,t\,e^{t}$$
$$g := 6\,t+e^{-t}\sin(4\,t)-7\,t\,e^{-3\,t} \quad (12)$$

```
> v1 := int(g*Wronskian([exp(t),t*exp(t)],t,determinant=true)[2]/Wronskian(
  [exp(-3*t),exp(t),t*exp(t)],t,determinant=true)[2],t);
```

$$v1 := \frac{(e^t)^3 t}{8} - \frac{(e^t)^3}{24} - \frac{e^{2t}\cos(4t)}{80} + \frac{e^{2t}\sin(4t)}{160} - \frac{e^{2t}\cos(2t)}{32} + \frac{e^{2t}\sin(2t)}{32} - \frac{e^{2t}(2\sin(2t) - 2\cos(2t))}{64} - \frac{7t^2}{32} \quad \textbf{(13)}$$

```
> v2 := int(-g*Wronskian([exp(-3*t),t*exp(t)],t,determinant=true)[2]
  /Wronskian([exp(-3*t),exp(t),t*exp(t)],t,determinant=true)[2],t);
```

$$v2 := \frac{3t^2}{2e^t} + \frac{27t}{8e^t} + \frac{27}{8e^t} - \frac{\left(-\frac{t}{5} - \frac{1}{25}\right)e^{-2t}\cos(4t)}{4} - \frac{\left(-\frac{t}{10} + \frac{3}{100}\right)e^{-2t}\sin(4t)}{4} + \frac{e^{-2t}\cos(4t)}{80} + \frac{e^{-2t}\sin(4t)}{160} + \frac{e^{-2t}\cos(2t)}{32} + \frac{e^{-2t}\sin(2t)}{32} + \frac{e^{-2t}(-2\sin(2t) - 2\cos(2t))}{64} - \frac{7t^2}{16(e^t)^4} - \frac{21t}{64(e^t)^4} - \frac{21}{256(e^t)^4} \quad \textbf{(14)}$$

```
> v3 := int(g*Wronskian([exp(-3*t),exp(t)],t,determinant=true)[2]/Wronskian(
  [exp(-3*t),exp(t),t*exp(t)],t,determinant=true)[2],t);
```

$$v3 := -\frac{3t}{2e^t} - \frac{3}{2e^t} - \frac{e^{-2t}\cos(4t)}{20} - \frac{e^{-2t}\sin(4t)}{40} - \frac{e^{-2t}\cos(2t)}{8} - \frac{e^{-2t}\sin(2t)}{8} - \frac{e^{-2t}(-2\sin(2t) - 2\cos(2t))}{16} + \frac{7t}{16(e^t)^4} + \frac{7}{64(e^t)^4} \quad \textbf{(15)}$$

```
> yp := simplify(yp);
```

$$yp := -\frac{7e^{-3t}\left(\left(-\frac{8\cos(4t)}{175} - \frac{4\sin(4t)}{175}\right)e^{2t} + \left(-\frac{64t}{7} - \frac{320}{21}\right)e^{3t} + t^2 + t + \frac{3}{8}\right)}{32} \quad \textbf{(16)}$$

Check it really is a solution (this is left hand side of our DE minus g -- it should be zero):

```
> simplify(diff(yp,t,t,t)+diff(yp,t,t)-5*diff(yp,t)+3*yp-(6*t+exp(-t)*sin(4*
  t)-7*t*exp(-3*t)));
```

$$0 \quad \textbf{(17)}$$

Once again, our general solution (which is exactly the same as before - if we simplified coefficients)...

```
> y = yp+yh;
```

$$y = -\frac{7e^{-3t}\left(\left(-\frac{8\cos(4t)}{175} - \frac{4\sin(4t)}{175}\right)e^{2t} + \left(-\frac{64t}{7} - \frac{320}{21}\right)e^{3t} + t^2 + t + \frac{3}{8}\right)}{32} + C1\,e^{-3t} + C2\,e^t + C3\,t\,e^t \quad \textbf{(18)}$$

Partial Fraction:

If s = d/dt, then our equation is L[y]=g and so (using very questionable notations): y = (1/L)g. We compute the partial fraction decomposition of 1/L.

```
> convert(1/(s^3+s^2-5*s+3),parfrac,s);
```

(19)

$$\frac{1}{16\,(s+3)} - \frac{1}{16\,(s-1)} + \frac{1}{4\,(s-1)^2} \quad (19)$$

Recall that $\frac{1}{(s-a)^k}[g] = (s-a)^{-k}[g] = e^{a \cdot t}\int\int \ldots \int e^{-a \cdot t} g$ (where that is a k-fold integral).

While it may not look like it, this is again an equivalent particular solution....

```
> yp :=simplify(  1/4*exp(t)*int(int(exp(-t)*g,t),t)
     -1/16*exp(t)*int(exp(-t)*g,t)
     +1/16*exp(-3*t)*int(exp(3*t)*g,t));
```

$$yp := -\frac{1}{32}\left(7\left(\left(\frac{2\cos(t)^2}{7} - \frac{2\cos(t)\sin(t)}{7} - \frac{4\sin(4t)}{175} - \frac{\cos(2t)}{7} - \frac{8\cos(4t)}{175} + \frac{\sin(2t)}{7} - \frac{1}{7}\right)e^{2t} + \left(-\frac{64t}{7} - \frac{320}{21}\right)e^{3t} + t^2 + t + \frac{3}{8}\right)e^{-3t}\right) \quad (20)$$

Check it really is a solution (this is left hand side of our DE minus g -- it should be zero):

```
> simplify(diff(yp,t,t,t)+diff(yp,t,t)-5*diff(yp,t)+3*yp-(6*t+exp(-t)*sin(4*
  t)-7*t*exp(-3*t)));
```

$$0 \quad (21)$$

Once again, our general solution....

```
> y = yh+yp;
```

$$y = C1\,e^{-3t} + C2\,e^{t} + C3\,t\,e^{t} - \frac{1}{32}\left(7\left(\left(\frac{2\cos(t)^2}{7} - \frac{2\cos(t)\sin(t)}{7} - \frac{4\sin(4t)}{175} - \frac{\cos(2t)}{7} - \frac{8\cos(4t)}{175} + \frac{\sin(2t)}{7} - \frac{1}{7}\right)e^{2t} + \left(-\frac{64t}{7} - \frac{320}{21}\right)e^{3t} + t^2 + t + \frac{3}{8}\right)e^{-3t}\right) \quad (22)$$

Converting to 1st order linear system.

We get the following system:

```
> diff(x1(t),t)=x2(t);
  diff(x2(t),t)=x3(t);
  diff(x3(t),t)=-3*x1(t)+5*x2(t)-x3(t)+g;
```

$$\frac{d}{dt}\,x1(t) = x2(t)$$

$$\frac{d}{dt}\,x2(t) = x3(t)$$

$$\frac{d}{dt}\,x3(t) = -3\,x1(t) + 5\,x2(t) - x3(t) + 6\,t + e^{-t}\sin(4\,t) - 7\,t\,e^{-3t} \quad (23)$$

```
> A := <<0,0,-3>|<1,0,5>|<0,1,-1>>;

  gvec := <<0,0,g>>;
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ -3 & 5 & -1 \end{bmatrix}$$

$$gvec := \begin{bmatrix} 0 \\ 0 \\ 6t + e^{-t}\sin(4t) - 7te^{-3t} \end{bmatrix} \quad \textbf{(24)}$$

Here I'll show off the system (this doesn't really do anything for us):

```
> x := t -> <<x1(t),x2(t),x3(t)>>:
  diff(x(t),t) = A.x(t)+gvec;
```

$$\begin{bmatrix} \frac{d}{dt} x1(t) \\ \frac{d}{dt} x2(t) \\ \frac{d}{dt} x3(t) \end{bmatrix} = \begin{bmatrix} x2(t) \\ x3(t) \\ -3\,x1(t) + 5\,x2(t) - x3(t) + 6t + e^{-t}\sin(4t) - 7te^{-3t} \end{bmatrix} \quad \textbf{(25)}$$

First, we will solve the homogeneous system x' = Ax. Since A is a constant matrix, this can be done with exp(At).

We could just make Maple compute the MatrixExponential for us - but where is the fun in that?

I'll compute the Jordan form and build up exp(At) from that.

Notice that A's characteristic polynomial is exactly the characteristic polynomial of our original differential equation. This will always be the case if we convert to a first order system in the way we did.

```
> factor(CharacteristicPolynomial(A,t));
```

$$(t+3)\,(t-1)^2 \quad \textbf{(26)}$$

We have eigenvalues -3 and 1. We need a single eigenvector with eigenvalue -3. I'll rescale Maple's answer so my eigenvector has integer entries (this isn't necessary):

```
> NullSpace(A+3*IdentityMatrix(3));
```

$$\left\{ \begin{bmatrix} \frac{1}{9} \\ -\frac{1}{3} \\ 1 \end{bmatrix} \right\} \quad \textbf{(27)}$$

```
> v1 := <<1,-3,9>>;
```

$$v1 := \begin{bmatrix} 1 \\ -3 \\ 9 \end{bmatrix} \quad \textbf{(28)}$$

We need a chain of length 2 for the eigenvalue 1.

We solve $(A-I)^2 x=0$ and need $(A-I)x \neq 0$. The first basis vector works:

```
> NullSpace((A-IdentityMatrix(3))^2);
```

$$\left\{\begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 2 \\ 1 \\ 0 \end{bmatrix}\right\} \tag{29}$$

```
> w2 := <<-1,0,1>>;
  w1 := (A-IdentityMatrix(3)).w2;
```

$$w2 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}$$

$$w1 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \tag{30}$$

```
> P := <v1|w1|w2>;
  P^(-1).A.P;
```

$$P := \begin{bmatrix} 1 & 1 & -1 \\ -3 & 1 & 0 \\ 9 & 1 & 1 \end{bmatrix}$$

$$\begin{bmatrix} -3 & 0 & 0 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{bmatrix} \tag{31}$$

Define our diagonal and nilponent parts:

```
> Diag := <<-3,0,0>|<0,1,0>|<0,0,1>>;
  N := <<0,0,0>|<0,0,0>|<0,1,0>>;
```

$$Diag := \begin{bmatrix} -3 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$N := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix} \tag{32}$$

$M = e^{At} = e^{PJP^{-1}t} = e^{P(Dt + Nt)P^{-1}} = Pe^{Dt}e^{Nt}P^{-1}$

Notice that $N^2=0$, so $e^{Nt} = I + Nt + 0 + 0 + ...$

Note: Once again, I will define this as a function of t so Maple will let me differentiate etc.

```
> M := P.<<exp(-3*t),0,0>|<0,exp(t),0>|<0,0,exp(t)>>.(IdentityMatrix(3)+N*t).
  P^(-1);
```

$$M := \begin{bmatrix} \frac{e^{-3t}}{16} + \frac{15\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4} & -\frac{e^{-3t}}{8} + \frac{e^{t}}{8} + \frac{e^{t}\, t}{2} & \frac{e^{-3t}}{16} - \frac{e^{t}}{16} + \frac{e^{t}\, t}{4} \\ -\frac{3\, e^{-3t}}{16} + \frac{3\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4} & \frac{3\, e^{-3t}}{8} + \frac{5\, e^{t}}{8} + \frac{e^{t}\, t}{2} & -\frac{3\, e^{-3t}}{16} + \frac{3\, e^{t}}{16} + \frac{e^{t}\, t}{4} \\ \frac{9\, e^{-3t}}{16} - \frac{9\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4} & -\frac{9\, e^{-3t}}{8} + \frac{9\, e^{t}}{8} + \frac{e^{t}\, t}{2} & \frac{9\, e^{-3t}}{16} + \frac{7\, e^{t}}{16} + \frac{e^{t}\, t}{4} \end{bmatrix} \quad \textbf{(33)}$$

This agrees with Maple:

```
> MatrixExponential(A*t);
```

$$\begin{bmatrix} \frac{(-12\,t+15)\, e^{-3t}\, e^{4t}}{16} + \frac{e^{-3t}}{16} & \frac{\left(4\, e^{4t}\, t + e^{4t} - 1\right) e^{-3t}}{8} & \frac{(4\,t-1)\, e^{-3t}\, e^{4t}}{16} + \frac{e^{-3t}}{16} \\ \frac{(-12\,t+3)\, e^{-3t}\, e^{4t}}{16} - \frac{3\, e^{-3t}}{16} & \frac{(4\,t+5)\, e^{-3t}\, e^{4t}}{8} + \frac{3\, e^{-3t}}{8} & \frac{(4\,t+3)\, e^{-3t}\, e^{4t}}{16} - \frac{3\, e^{-3t}}{16} \\ \frac{(-12\,t-9)\, e^{-3t}\, e^{4t}}{16} + \frac{9\, e^{-3t}}{16} & \frac{(4\,t+9)\, e^{-3t}\, e^{4t}}{8} - \frac{9\, e^{-3t}}{8} & \frac{(4\,t+7)\, e^{-3t}\, e^{4t}}{16} + \frac{9\, e^{-3t}}{16} \end{bmatrix} \quad \textbf{(34)}$$

Our homogeneous solution:

```
> xh := M.<<C1,C2,C3>>;
```

$$xh := \begin{bmatrix} \left(\frac{e^{-3t}}{16} + \frac{15\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4}\right) C1 + \left(-\frac{e^{-3t}}{8} + \frac{e^{t}}{8} + \frac{e^{t}\, t}{2}\right) C2 + \left(\frac{e^{-3t}}{16} - \frac{e^{t}}{16} + \frac{e^{t}\, t}{4}\right) C3 \\ \left(-\frac{3\, e^{-3t}}{16} + \frac{3\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4}\right) C1 + \left(\frac{3\, e^{-3t}}{8} + \frac{5\, e^{t}}{8} + \frac{e^{t}\, t}{2}\right) C2 + \left(-\frac{3\, e^{-3t}}{16} + \frac{3\, e^{t}}{16} + \frac{e^{t}\, t}{4}\right) C3 \\ \left(\frac{9\, e^{-3t}}{16} - \frac{9\, e^{t}}{16} - \frac{3\, e^{t}\, t}{4}\right) C1 + \left(-\frac{9\, e^{-3t}}{8} + \frac{9\, e^{t}}{8} + \frac{e^{t}\, t}{2}\right) C2 + \left(\frac{9\, e^{-3t}}{16} + \frac{7\, e^{t}}{16} + \frac{e^{t}\, t}{4}\right) C3 \end{bmatrix} \quad \textbf{(35)}$$

Note: The top row is a general solution of our original companion equation. This looks way more complicated since the C1, C2, C3 constants are different from the original choice. In fact, while this formula looks worse than what we got before, it is nicer to work with when dealing with initial conditions C1 corresponds with y(0), C2 with y'(0), and C2 with y''(0).

Maple note: The "map" command allows us to apply a procedure to every entry of something. Here I am substituting t=0 and simplifying every entry of xh.

```
> <<yAt0,DyAt0,DDyAt0>> = map(X -> simplify(subs(t=0,X)), xh);
```

$$\begin{bmatrix} yAt0 \\ DyAt0 \\ DDyAt0 \end{bmatrix} = \begin{bmatrix} C1 \\ C2 \\ C3 \end{bmatrix} \quad \textbf{(36)}$$

Now for our system version of variation of parameters. We have $x_p(t) = M(t)\int M(t)^{-1} g$

Note: We use the map command to integrate each entry of $M(t)^{-1} g(t)$.

```
> xp := M.map(f -> simplify(int(f,t)),M^(-1).gvec);
```

$$xp := \left[\left[\frac{1}{2}\left(3\left(\frac{e^{-3t}}{16}+\frac{15\,e^{t}}{16}-\frac{3\,e^{t}\,t}{4}\right)e^{-4t}\left(\left(\left(\left(\frac{t}{30}+\frac{3}{200}\right)\cos(4\,t)+\frac{\sin(4\,t)\left(t-\frac{1}{20}\right)}{60}\right)e^{2t}\right.\right.\right.\right.$$

$$-\frac{e^{6t}\cos(4\,t)}{120}+\left(t^2+\frac{9}{4}\,t+\frac{9}{4}\right)e^{3t}+\frac{e^{6t}\sin(4\,t)}{240}+\left(\frac{t}{12}-\frac{1}{36}\right)e^{7t}-\frac{7\,t^2\,e^{4t}}{48}-\frac{7\,t^2}{24}-\frac{7\,t}{32}$$

$$-\frac{7}{128}\Bigg)\Bigg)+\frac{1}{2}\left(3\left(-\frac{e^{-3t}}{8}+\frac{e^{t}}{8}+\frac{e^{t}\,t}{2}\right)\left(\left(\left(\left(\frac{t}{30}-\frac{11}{600}\right)\cos(4\,t)+\frac{\left(t-\frac{21}{20}\right)\sin(4\,t)}{60}\right)e^{2t}\right.\right.\right.$$

$$+\frac{e^{6t}\cos(4\,t)}{40}+\left(t^2+\frac{5}{4}\,t+\frac{5}{4}\right)e^{3t}-\frac{e^{6t}\sin(4\,t)}{80}+\left(-\frac{t}{4}+\frac{1}{12}\right)e^{7t}+\frac{7\,t^2\,e^{4t}}{16}-\frac{7\,t^2}{24}+\frac{7\,t}{96}$$

$$+\frac{7}{384}\Bigg)e^{-4t}\Bigg)+\frac{1}{2}\left(3\left(\frac{e^{-3t}}{16}-\frac{e^{t}}{16}+\frac{e^{t}\,t}{4}\right)e^{-4t}\left(\left(\left(\left(\frac{t}{30}-\frac{31}{600}\right)\cos(4\,t)\right.\right.\right.\right.$$

$$+\frac{\sin(4\,t)\left(t-\frac{41}{20}\right)}{60}\Bigg)e^{2t}-\frac{3\,e^{6t}\cos(4\,t)}{40}+\left(t^2+\frac{1}{4}\,t+\frac{1}{4}\right)e^{3t}+\frac{3\,e^{6t}\sin(4\,t)}{80}+\left(\frac{3\,t}{4}\right.$$

$$\left.-\frac{1}{4}\right)e^{7t}-\frac{21\,t^2\,e^{4t}}{16}-\frac{7\,t^2}{24}+\frac{35\,t}{96}+\frac{35}{384}\Bigg)\Bigg)\Bigg],$$

$$\left[\frac{1}{2}\left(3\left(-\frac{3\,e^{-3t}}{16}+\frac{3\,e^{t}}{16}-\frac{3\,e^{t}\,t}{4}\right)e^{-4t}\left(\left(\left(\left(\frac{t}{30}+\frac{3}{200}\right)\cos(4\,t)+\frac{\sin(4\,t)\left(t-\frac{1}{20}\right)}{60}\right)e^{2t}\right.\right.\right.\right.$$

$$-\frac{e^{6t}\cos(4\,t)}{120}+\left(t^2+\frac{9}{4}\,t+\frac{9}{4}\right)e^{3t}+\frac{e^{6t}\sin(4\,t)}{240}+\left(\frac{t}{12}-\frac{1}{36}\right)e^{7t}-\frac{7\,t^2\,e^{4t}}{48}-\frac{7\,t^2}{24}-\frac{7\,t}{32}$$

$$-\frac{7}{128}\Bigg)\Bigg)+\frac{1}{2}\left(3\left(\frac{3\,e^{-3t}}{8}+\frac{5\,e^{t}}{8}+\frac{e^{t}\,t}{2}\right)\left(\left(\left(\left(\frac{t}{30}-\frac{11}{600}\right)\cos(4\,t)\right.\right.\right.\right.$$

$$+\frac{\left(t-\frac{21}{20}\right)\sin(4\,t)}{60}\Bigg)e^{2t}+\frac{e^{6t}\cos(4\,t)}{40}+\left(t^2+\frac{5}{4}\,t+\frac{5}{4}\right)e^{3t}-\frac{e^{6t}\sin(4\,t)}{80}+\left(-\frac{t}{4}+\frac{1}{12}\right)e^{7t}$$

(37)

$$+\frac{7\,t^2\,e^{4\,t}}{16}-\frac{7\,t^2}{24}+\frac{7\,t}{96}+\frac{7}{384}\Big)\,e^{-4\,t}\Big)+\frac{1}{2}\Big(3\Big(-\frac{3\,e^{-3\,t}}{16}+\frac{3\,e^{t}}{16}+\frac{e^{t}\,t}{4}\Big)\,e^{-4\,t}\Big(\Big(\Big(\frac{t}{30}$$

$$-\frac{31}{600}\Big)\cos(4\,t)+\frac{\sin(4\,t)\,\big(t-\frac{41}{20}\big)}{60}\Big)\,e^{2\,t}-\frac{3\,e^{6\,t}\cos(4\,t)}{40}+\Big(t^2+\frac{1}{4}\,t+\frac{1}{4}\Big)\,e^{3\,t}$$

$$+\frac{3\,e^{6\,t}\sin(4\,t)}{80}+\Big(\frac{3\,t}{4}-\frac{1}{4}\Big)\,e^{7\,t}-\frac{21\,t^2\,e^{4\,t}}{16}-\frac{7\,t^2}{24}+\frac{35\,t}{96}+\frac{35}{384}\Big)\Big)\Big]\,,$$

$$\Big[\frac{1}{2}\Big(3\Big(\frac{9\,e^{-3\,t}}{16}-\frac{9\,e^{t}}{16}-\frac{3\,e^{t}\,t}{4}\Big)\,e^{-4\,t}\Big(\Big(\Big(\frac{t}{30}+\frac{3}{200}\Big)\cos(4\,t)+\frac{\sin(4\,t)\,\big(t-\frac{1}{20}\big)}{60}\Big)\,e^{2\,t}$$

$$-\frac{e^{6\,t}\cos(4\,t)}{120}+\Big(t^2+\frac{9}{4}\,t+\frac{9}{4}\Big)\,e^{3\,t}+\frac{e^{6\,t}\sin(4\,t)}{240}+\Big(\frac{t}{12}-\frac{1}{36}\Big)\,e^{7\,t}-\frac{7\,t^2\,e^{4\,t}}{48}-\frac{7\,t^2}{24}-\frac{7\,t}{32}$$

$$-\frac{7}{128}\Big)\Big)+\frac{1}{2}\Big(3\Big(-\frac{9\,e^{-3\,t}}{8}+\frac{9\,e^{t}}{8}+\frac{e^{t}\,t}{2}\Big)\Big(\Big(\Big(\frac{t}{30}-\frac{11}{600}\Big)\cos(4\,t)$$

$$+\frac{\big(t-\frac{21}{20}\big)\sin(4\,t)}{60}\Big)\,e^{2\,t}+\frac{e^{6\,t}\cos(4\,t)}{40}+\Big(t^2+\frac{5}{4}\,t+\frac{5}{4}\Big)\,e^{3\,t}-\frac{e^{6\,t}\sin(4\,t)}{80}+\Big(-\frac{t}{4}+\frac{1}{12}\Big)\,e^{7\,t}$$

$$+\frac{7\,t^2\,e^{4\,t}}{16}-\frac{7\,t^2}{24}+\frac{7\,t}{96}+\frac{7}{384}\Big)\,e^{-4\,t}\Big)+\frac{1}{2}\Big(3\Big(\frac{9\,e^{-3\,t}}{16}+\frac{7\,e^{t}}{16}+\frac{e^{t}\,t}{4}\Big)\,e^{-4\,t}\Big(\Big(\Big(\frac{t}{30}$$

$$-\frac{31}{600}\Big)\cos(4\,t)+\frac{\sin(4\,t)\,\big(t-\frac{41}{20}\big)}{60}\Big)\,e^{2\,t}-\frac{3\,e^{6\,t}\cos(4\,t)}{40}+\Big(t^2+\frac{1}{4}\,t+\frac{1}{4}\Big)\,e^{3\,t}$$

$$+\frac{3\,e^{6\,t}\sin(4\,t)}{80}+\Big(\frac{3\,t}{4}-\frac{1}{4}\Big)\,e^{7\,t}-\frac{21\,t^2\,e^{4\,t}}{16}-\frac{7\,t^2}{24}+\frac{35\,t}{96}+\frac{35}{384}\Big)\Big)\Big]\Big]$$

Thus a general solution of our first order linear system is...

```
> x = xh+xp;
```

$$x=\Big[\Big[\Big(\frac{e^{-3\,t}}{16}+\frac{15\,e^{t}}{16}-\frac{3\,e^{t}\,t}{4}\Big)\,C1+\Big(-\frac{e^{-3\,t}}{8}+\frac{e^{t}}{8}+\frac{e^{t}\,t}{2}\Big)\,C2+\Big(\frac{e^{-3\,t}}{16}-\frac{e^{t}}{16}+\frac{e^{t}\,t}{4}\Big)\,C3 \tag{38}$$

$$+\frac{1}{2}\Big(3\Big(\frac{e^{-3\,t}}{16}+\frac{15\,e^{t}}{16}-\frac{3\,e^{t}\,t}{4}\Big)\,e^{-4\,t}\Big(\Big(\Big(\frac{t}{30}+\frac{3}{200}\Big)\cos(4\,t)+\frac{\sin(4\,t)\,\big(t-\frac{1}{20}\big)}{60}\Big)\,e^{2\,t}$$

$$-\frac{e^{6t}\cos(4t)}{120}+\left(t^2+\frac{9}{4}t+\frac{9}{4}\right)e^{3t}+\frac{e^{6t}\sin(4t)}{240}+\left(\frac{t}{12}-\frac{1}{36}\right)e^{7t}-\frac{7t^2e^{4t}}{48}-\frac{7t^2}{24}-\frac{7t}{32}$$

$$-\frac{7}{128}\Big)\Big)+\frac{1}{2}\Big(3\left(-\frac{e^{-3t}}{8}+\frac{e^{t}}{8}+\frac{e^{t}t}{2}\right)\Big(\Big(\left(\frac{t}{30}-\frac{11}{600}\right)\cos(4t)+\frac{\left(t-\frac{21}{20}\right)\sin(4t)}{60}\Big)e^{2t}$$

$$+\frac{e^{6t}\cos(4t)}{40}+\left(t^2+\frac{5}{4}t+\frac{5}{4}\right)e^{3t}-\frac{e^{6t}\sin(4t)}{80}+\left(-\frac{t}{4}+\frac{1}{12}\right)e^{7t}+\frac{7t^2e^{4t}}{16}-\frac{7t^2}{24}+\frac{7t}{96}$$

$$+\frac{7}{384}\Big)e^{-4t}\Big)+\frac{1}{2}\Big(3\left(\frac{e^{-3t}}{16}-\frac{e^{t}}{16}+\frac{e^{t}t}{4}\right)e^{-4t}\Big(\Big(\left(\frac{t}{30}-\frac{31}{600}\right)\cos(4t)$$

$$+\frac{\sin(4t)\left(t-\frac{41}{20}\right)}{60}\Big)e^{2t}-\frac{3e^{6t}\cos(4t)}{40}+\left(t^2+\frac{1}{4}t+\frac{1}{4}\right)e^{3t}+\frac{3e^{6t}\sin(4t)}{80}+\Big(\frac{3t}{4}$$

$$-\frac{1}{4}\Big)e^{7t}-\frac{21t^2e^{4t}}{16}-\frac{7t^2}{24}+\frac{35t}{96}+\frac{35}{384}\Big)\Big)\Big]\,,$$

$$\Big[\left(-\frac{3e^{-3t}}{16}+\frac{3e^{t}}{16}-\frac{3e^{t}t}{4}\right)C1+\left(\frac{3e^{-3t}}{8}+\frac{5e^{t}}{8}+\frac{e^{t}t}{2}\right)C2+\left(-\frac{3e^{-3t}}{16}+\frac{3e^{t}}{16}+\frac{e^{t}t}{4}\right)C3$$

$$+\frac{1}{2}\Big(3\left(-\frac{3e^{-3t}}{16}+\frac{3e^{t}}{16}-\frac{3e^{t}t}{4}\right)e^{-4t}\Big(\Big(\left(\frac{t}{30}+\frac{3}{200}\right)\cos(4t)+\frac{\sin(4t)\left(t-\frac{1}{20}\right)}{60}\Big)e^{2t}$$

$$-\frac{e^{6t}\cos(4t)}{120}+\left(t^2+\frac{9}{4}t+\frac{9}{4}\right)e^{3t}+\frac{e^{6t}\sin(4t)}{240}+\left(\frac{t}{12}-\frac{1}{36}\right)e^{7t}-\frac{7t^2e^{4t}}{48}-\frac{7t^2}{24}-\frac{7t}{32}$$

$$-\frac{7}{128}\Big)\Big)+\frac{1}{2}\Big(3\left(\frac{3e^{-3t}}{8}+\frac{5e^{t}}{8}+\frac{e^{t}t}{2}\right)\Big(\Big(\left(\frac{t}{30}-\frac{11}{600}\right)\cos(4t)$$

$$+\frac{\left(t-\frac{21}{20}\right)\sin(4t)}{60}\Big)e^{2t}+\frac{e^{6t}\cos(4t)}{40}+\left(t^2+\frac{5}{4}t+\frac{5}{4}\right)e^{3t}-\frac{e^{6t}\sin(4t)}{80}+\left(-\frac{t}{4}+\frac{1}{12}\right)e^{7t}$$

$$+\frac{7t^2e^{4t}}{16}-\frac{7t^2}{24}+\frac{7t}{96}+\frac{7}{384}\Big)e^{-4t}\Big)+\frac{1}{2}\Big(3\left(-\frac{3e^{-3t}}{16}+\frac{3e^{t}}{16}+\frac{e^{t}t}{4}\right)e^{-4t}\Big(\Big(\Big(\frac{t}{30}$$

$$- \frac{31}{600}\Big) \cos(4\,t) + \frac{\sin(4\,t)\left(t - \frac{41}{20}\right)}{60}\Bigg) e^{2\,t} - \frac{3\,e^{6\,t}\cos(4\,t)}{40} + \left(t^2 + \frac{1}{4}\,t + \frac{1}{4}\right) e^{3\,t}$$

$$+ \frac{3\,e^{6\,t}\sin(4\,t)}{80} + \left(\frac{3\,t}{4} - \frac{1}{4}\right) e^{7\,t} - \frac{21\,t^2\,e^{4\,t}}{16} - \frac{7\,t^2}{24} + \frac{35\,t}{96} + \frac{35}{384}\Bigg)\Bigg)\Bigg],$$

$$\Bigg[\left(\frac{9\,e^{-3\,t}}{16} - \frac{9\,e^{t}}{16} - \frac{3\,e^{t}\,t}{4}\right) C1 + \left(-\frac{9\,e^{-3\,t}}{8} + \frac{9\,e^{t}}{8} + \frac{e^{t}\,t}{2}\right) C2 + \left(\frac{9\,e^{-3\,t}}{16} + \frac{7\,e^{t}}{16} + \frac{e^{t}\,t}{4}\right) C3$$

$$+ \frac{1}{2}\Bigg(3\left(\frac{9\,e^{-3\,t}}{16} - \frac{9\,e^{t}}{16} - \frac{3\,e^{t}\,t}{4}\right) e^{-4\,t}\Bigg(\Bigg(\Bigg(\left(\frac{t}{30} + \frac{3}{200}\right)\cos(4\,t) + \frac{\sin(4\,t)\left(t - \frac{1}{20}\right)}{60}\Bigg) e^{2\,t}$$

$$- \frac{e^{6\,t}\cos(4\,t)}{120} + \left(t^2 + \frac{9}{4}\,t + \frac{9}{4}\right) e^{3\,t} + \frac{e^{6\,t}\sin(4\,t)}{240} + \left(\frac{t}{12} - \frac{1}{36}\right) e^{7\,t} - \frac{7\,t^2\,e^{4\,t}}{48} - \frac{7\,t^2}{24} - \frac{7\,t}{32}$$

$$- \frac{7}{128}\Bigg)\Bigg)\Bigg) + \frac{1}{2}\Bigg(3\left(-\frac{9\,e^{-3\,t}}{8} + \frac{9\,e^{t}}{8} + \frac{e^{t}\,t}{2}\right)\Bigg(\Bigg(\Bigg(\left(\frac{t}{30} - \frac{11}{600}\right)\cos(4\,t)$$

$$+ \frac{\left(t - \frac{21}{20}\right)\sin(4\,t)}{60}\Bigg) e^{2\,t} + \frac{e^{6\,t}\cos(4\,t)}{40} + \left(t^2 + \frac{5}{4}\,t + \frac{5}{4}\right) e^{3\,t} - \frac{e^{6\,t}\sin(4\,t)}{80} + \left(-\frac{t}{4} + \frac{1}{12}\right) e^{7\,t}$$

$$+ \frac{7\,t^2\,e^{4\,t}}{16} - \frac{7\,t^2}{24} + \frac{7\,t}{96} + \frac{7}{384}\Bigg) e^{-4\,t}\Bigg)\Bigg) + \frac{1}{2}\Bigg(3\left(\frac{9\,e^{-3\,t}}{16} + \frac{7\,e^{t}}{16} + \frac{e^{t}\,t}{4}\right) e^{-4\,t}\Bigg(\Bigg(\Bigg(\left(\frac{t}{30}\right.$$

$$- \frac{31}{600}\Big) \cos(4\,t) + \frac{\sin(4\,t)\left(t - \frac{41}{20}\right)}{60}\Bigg) e^{2\,t} - \frac{3\,e^{6\,t}\cos(4\,t)}{40} + \left(t^2 + \frac{1}{4}\,t + \frac{1}{4}\right) e^{3\,t}$$

$$+ \frac{3\,e^{6\,t}\sin(4\,t)}{80} + \left(\frac{3\,t}{4} - \frac{1}{4}\right) e^{7\,t} - \frac{21\,t^2\,e^{4\,t}}{16} - \frac{7\,t^2}{24} + \frac{35\,t}{96} + \frac{35}{384}\Bigg)\Bigg)\Bigg]\Bigg]$$

The top row is then our "y" (a solution of the original equation).

Note: Maple has stored (xh+xp) as a 3x1 matrix. Thus [1,1] picks off the (1,1)-entry (i.e., the top row).

```
> ySoln := simplify((xh+xp)[1,1]);
```

$$ySoln := -\frac{1}{32}\Bigg(7\Bigg(\Bigg(\Bigg(\left(\frac{24\,C1}{7} - \frac{16\,C2}{7} - \frac{8\,C3}{7}\right) t - \frac{30\,C1}{7} - \frac{4\,C2}{7} + \frac{2\,C3}{7}\Bigg) e^{4\,t} + \Bigg(-\frac{8\cos(4\,t)}{175} - \frac{4\sin(4\,t)}{175}\Bigg) e^{2\,t} + \left(-\frac{64\,t}{7} - \frac{320}{21}\right) e^{3\,t} + t^2 + t - \frac{2\,C1}{7} + \frac{4\,C2}{7} - \frac{2\,C3}{7} + \frac{3}{8}\Bigg) e^{-3\,t}\Bigg) \qquad (39)$$

This looks quite different from our previous general solutions. Let's make sure it actually works (i.e., plug it into the DE's left hand side minus g(t)) -- we should get 0 here:

```
> simplify(diff(ySoln,t,t,t)+diff(ySoln,t,t)-5*diff(ySoln,t)+3*ySoln-(6*t+exp
  (-t)*sin(4*t)-7*t*exp(-3*t)));
```

$$0 \qquad (40)$$

Or Make Maple Do It: dsolve

Obiviously, the easiest way is just to make Maple solve it for us...

```
> dsolve(diff(y(t),t,t,t)+diff(y(t),t,t)-5*diff(y(t),t)+3*y(t)=6*t+exp(-t)*
  sin(4*t)-7*t*exp(-3*t));
```

$$y(t) = -\frac{e^{-3t}\left(-192\,e^{2t}\cos(4t) - 96\,e^{2t}\sin(4t) - 38400\,e^{3t}t + 4200\,t^2 - 64000\,e^{3t} + 4200\,t + 1575\right)}{19200} + \_C1\,e^{t} + \_C2\,e^{-3t} + \_C3\,e^{t}\,t \qquad (41)$$